

Chapitre 6

Instruments à taux fixe

A. Introduction	111
B. La fonction FluxTF	111
C. La fonction CouponCouruTF	115
D. La fonction TauxRendementTF	118
E. La fonction PrixPleinTF	122
F. La fonction PrixPiedTF	124
G. La fonction SpreadSurCourbeTF	126
H. La fonction PrixPleinSurCourbeTF	130
I. Les fonctions de gestion du risque d'un instrument à taux fixe	133

A. Introduction

Vous allez dans ce chapitre réaliser un ensemble de fonctions utiles pour la gestion des instruments à taux fixe.



Un instrument à taux fixe paye à intervalles réguliers des intérêts basés sur le même taux tout au long de sa vie. Le montant des intérêts peut ne pas être constant.

Par exemple les obligations de l'Etat français, les OAT, sont des instruments à taux fixe. Pour plus d'informations sur les obligations, vous pouvez consulter l'article qui leur est consacré sur Wikipédia, à l'adresse : [http://fr.wikipedia.org/wiki/Obligation_\(finance\)](http://fr.wikipedia.org/wiki/Obligation_(finance))

Nous ne considérons que les instruments à remboursement in fine, c'est-à-dire dont le flux de capital est unique et à la date de maturité de l'instrument.

B. La fonction FluxTF

1. Objectif

Cette fonction détermine les dates et les flux d'un instrument à taux fixe entre la date de calcul et la date de maturité.

2. Arguments

Nom de l'argument	Description	Format de l'argument
DateDeCalcul	Date de calcul	Date
DateDeMaturite	Date de maturité de l'instrument	Date
dblCoupon	Coupon (%)	Double, coupon de l'instrument
iFrequence	Fréquence	Entier, nombre de coupon par an de l'instrument
Base	Base	Variant
dblValeurRemboursement	Valeur de remboursement	Double
ModeAjustement	Mode d'ajustement (optionnel)	Variant
bPremierCouponPlein	Premier Coupon Plein (optionnel)	Booléen
TypeCouponBrise	Type coupon brisé (optionnel)	Variant
DateDeDepart	Date de départ de l'instrument (optionnel, obligatoire si le type de coupon brisé est renseigné)	Date

3. Mode de calcul

Si les flux ne sont pas ajustés selon leur durée, les paiements d'intérêts sont tous égaux :

$Flux(i)=dblValeurRemboursement * Coupon / iFrequence$

Pour un investissement de 100 € recevant un coupon de 5% par an, payé en 2 fois, les flux d'intérêts seront de 2.5 €.

Si les flux sont ajustés selon leur durée, par exemple en mode Modified Following, il faut pondérer chaque flux par la fonction FractionAnnee :

$dblFrac=FractionAnnee(DateFlux(i-1);DateFlux(i);ModeAjustement)$

$Flux(i) = dblValeurRemboursement * Coupon * dblFrac$

Que les flux d'intérêts soient ajustés ou non, le dernier flux de l'instrument est égal au flux d'intérêts plus le flux de remboursement.

4. Code de la fonction FluxTF

Pour déterminer la date des flux nous utilisons la fonction DatesDesFlux.

```
Option Base 1
Option Explicit

Function FluxTF(DateDeCalcul As Date, DateDeMaturite As Date, _
    dblCoupon As Double, iFrequence As Integer, _
    Base As Variant, dblValeurRemboursement _
    As Double, Optional ModeAjustement As Variant = 0, _
    Optional bPremierCouponPlein As Boolean = False, _
    Optional TypeCouponBrise As Variant = 0, _
    Optional DateDeDepart As Date = 0)

'////////////////////
' Cette fonction détermine les flux d'un
' instrument à taux fixe
'////////////////////
Dim TabDateFlux           'Tableau renvoyé par la fonction DatesDesFlux
Dim dblFrac As Double     'Fraction année
Dim TabSortie             'Données renvoyées
Dim i As Integer          'Variable de boucle
Dim iFreq As Integer      'Variable pour gérer le cas des zéro-coupon

'Si la fréquence est égale à 0
' alors c'est un instrument zéro coupon
' Pour que les formules ci-après fonctionnent
' il faut utiliser une seconde variable pour
' conserver la fréquence
If iFrequence = 0 Then
    iFreq = 1
Else
    iFreq = iFrequence
End If

'Test de la validité des données
If TypeCouponBrise <> 0 And DateDeDepart = 0 Then Exit Function

'Détermine les dates de flux
TabDateFlux = DatesDesFlux(DateDeCalcul, DateDeMaturite, iFrequence,
    _
    ModeAjustement, TypeCouponBrise, DateDeDepart)
'Dimensionne le tableau de sortie
ReDim TabSortie(2, UBound(TabDateFlux))

'Traitement du premier flux
```

```

'La date du premier flux
TabSortie(1, 1) = TabDateFlux(1)

'Test coupon brisé non encore échu
If TypeCouponBrise <> 0 And bPremierCouponPlein = False And _
    TabDateFlux(0) = DateDeDepart Then
    'flux pondéré par sa durée
    dblFrac = FractionAnnee((TabDateFlux(0)), (TabDateFlux(1)), Base)
    TabSortie(2, 1) = dblValeurRemboursement * dblCoupon * dblFrac
Else
    'flux plein
    TabSortie(2, 1) = dblValeurRemboursement * dblCoupon / iFreq
End If

'Les autres flux
If UBound(TabDateFlux) > 1 Then
    'S'il y a d'autres flux, on boucle sur le tableau de dates
    For i = 2 To UBound(TabDateFlux)
        'La date du flux
        TabSortie(1, i) = TabDateFlux(i)
        'Calcul du flux selon le mode d'ajustement
        If ModeAjustement <> 0 Then
            dblFrac = FractionAnnee((TabDateFlux(i - 1)), _
                (TabDateFlux(i)), Base)
            TabSortie(2, i) = dblValeurRemboursement * _
                dblCoupon * dblFrac
        Else
            'S'il n'y a pas de mode d'ajustement
            TabSortie(2, i) = dblValeurRemboursement * dblCoupon /
iFreq
        End If
    Next
End If

'Au dernier flux on ajoute le flux de capital
' (instrument in fine)
TabSortie(2, UBound(TabDateFlux)) = TabSortie(2, UBound(TabDateFlux)) _
    + dblValeurRemboursement

FluxTF = TabSortie
End Function

```